
ABSTRACT

A radar or star chart graphically shows the size of the target numeric variable among categories. On a radar chart, the chart
statistics are displayed along spokes from the center of the chart. The GRADAR procedure in SAS/GRAPH 9.1 creates radar
chart. The GRADAR procedure provides five star types with interactive features.

The interactive features are:

� By an ActiveX control – (pop-up data tips, drill-down links, and interactive menus)
� By a Metaview applet – (data tips, drill-down links, or some interactivity such as zooming, panning, and slide shows).
� By Static Images – (data tips, and drill-down links, or animation).

The hypothetical clinical data used throughout this paper for illustration purpose are:

� On-therapy Adverse Events
� On-therapy Serious Adverse Events
� QTC Change from Baseline
� Clinical Laboratory Data with Worst Case in High Direction
� ABPM Data

The SAS products utilized in this paper are SAS BASE� and SAS/GRAPH 9.1 on a PC Windows platform.

INTRODUCTION

The procedure GRADAR is a new graphical procedure provided in SAS/GRAPH 9.1 for creating radar charts. A radar chart is
a type of graphical presentation using spokes to illustrate the size of the target numeric variable among categories. On a
radar chart, the chart statistics are displayed along spokes from the center of the chart. The radar charts are called in different
names, such as star charts, polar chart, spider plots, or cobweb plots, because the radar chart has different types, such as
star types of corona, polygon, radial, spoke, and wedge provided from the Procedure GRADAR. These new types of data
displays provide the novel look of graphical presentations including interactive features. Figure 1 illustrates the star type that
is provided from the procedure.

Figure 1. Star Types

The procedure GRADAR supports the following device drivers: 1) ACTIVEX, 2) ACTXIMG, 3) GIF, JPEG, PNG, 4) GIFANIM,
and 5) JAVAMETA. Some of these device drivers offer many interactive features, such as ODS styles, pop-up data tips, drill-
down links, interactive menu, slide shows, and animation. Table 1 shows the interactive features supported by device driver.
The technologies used in the interactive features are: ActiveX control (Microsoft®) and Metaview Applet (Sun Microsystems®).

Device Driver Features Supported
ACTXIMG ODS styles, pop-up data tips, and drill-down links, static graphics with no interactivity.
ACTIVEX ODS styles, pop-up data tips, drill-down links, interactivity via pop-up menus
GIF,JPEG, PNG Pop-up data tips, drill-down links, static graphics with no interactivity
GIFANIM Slide show of static images with no interactivity
JAVAMETA Pop-up data tips, drill-down links, some interactivity such as zooming and slide shows.

Table 1. Interactive Features Supported by Device Driver

What does your audience require to view the presentation in addition to the browser? Table 2 demonstrates the additional
requirements needed for a presentation generated with the ACTIVEX driver and the JAVAMETA driver.

Device Driver Additional Requirements
ACTXIMG None
ACTIVEX The presentation must be viewed on a Windows system with the SAS ActiveX control

installed locally.
GIF,JPEG, PNG None
GIFANIM None
JAVAMETA The JAVA applet files must be installed locally or accessible by the client machine. The

JAVA plug-in is not required.

Table 2. Additional Requirements Needed to View the Presentation

The traditional clinical data displays are from the procedures of GPLOT, GCHART, … etc. This paper uses radar charts for
the clinical data display. The data used throughout of the paper are from a hypothetic clinical trial and are for illustration
purposes. This is a new data display application with a novel look at the clinical data. Five types of clinical data are used: 1)
clinical adverse events (AE), and serious AE, 2) clinical laboratory test data with worst case in the high direction, 3) corrected
QT (QTc) interval data from electrocardiogram (ECG) measures, and 4) ambulatory blood pressure measurement (ABPM)
data.

Some sample code is provided in the paper.

STEPS TO CREATE A RADAR CHART AND RADAR CHART LIMITATIONS

The steps to create a radar chart are described as follows:

� Select Spoke Variable (Category or Chronicle Variable)
� Standardize Frequency Count Definitions
� Select Star Type, Device Driver, and Interactive Features
� Prepare Data
� Call GRADAR Procedure
� Display and Interpret the Results

The GRADAR procedure poses limitations on star label, number of valid observations, number of vertices, and data tips
characters.

� Maximum characters for Star Label: 32
� Minimum number of Valid Observations: 3
� Maximum number of vertices: 360
� Maximum characters for data tips: 1024

WEDGE TYPE OF AE DATA DISPLAY

In clinical trial studies, the investigators are responsible for recording the subjects AE at each clinical office visit or
assessment. An adverse experience is defined as: any noxious, pathological, or unintended change in anatomical,
physiological, or metabolic functions as indicated by physical signs, symptoms, and/or laboratory, changes
occurring in any phase of the clinical study whether associated with the study drug, active comparator, or placebo,
and whether or not considered drug related. In other words, an AE is some unplanned, unwanted event which
occurs to a subject and which is possible related to the study medication and therapy.

AE data are important safety information in the clinical trials. The following AE summary table is a sample display
of AE report.

Summary of All Adverse Events

Body System Dose A Dose B
Preferred Term (N=314) (N=326)

ANY EVENT 283 (90%) 274 (84%)

EAR NOSE & THROAT
Any Event 82 (26%) 90 (28%)
Upper respiratory tract infection 39 (12%) 37 (11%)
Throat irritation 11 (4%) 11 (3%)
Upper respiratory inflammation 5 (2%) 10 (3%)
Nasal congestion/blockage 5 (2%) 6 (2%)
Rhinitis 5 (2%) 5 (2%)
Sinusitis 5 (2%) 4 (1%)
Rhinorrhea/post nasal drip 4 (1%) 1 (<1%)
Hoarseness/dysphonia 1 (<1%) 3 (<1%)
Ear nose & throat signs & symptoms 0 3 (<1%)
Epistaxis 1 (<1%) 2 (<1%)
Ear signs & symptoms 1 (<1%) 1 (<1%)

Table 3. Sample AE Report

The AE summary table displays either overall on-therapy AE counts or a subset of AE of interest. The subset of
AE may focus on fatal AE, serious AE, AE intensity, AE relationship to study medication, or study specific AE of
interest.

The graphical presentation of AE information, shown on summary table, provides visual comparison. In some
cases, it may provide safety signals detection. Gastrointestinal (GI) AE are selected for graphical radar chart
application. GI AE of concern are: Nausea, Abdominal Pain, Diarrhea, and Vomiting.

libname sb 'c:\gradar';

data mstone;
set sb.mstone;
sdur=ltdat - ftdat +1;
count=1;

keep pid rand race sex age sdur count;
proc sort;by rand;
proc summary data=mstone;
by rand;
var count;
output out=p0(keep=rand total) sum=total;

data ae;
set sb.ae;

if giae_c='Y' and ses_id=1;
count=1;
keep rand pid body1 pref body1_c pref_c count;
run;

proc sort data=ae nodupkey;by pid pref_c;
proc sort data=ae;by rand pref;

proc summary;
var count;
by rand pref;
output out=p1 sum=sae;
run;
data final;
merge p0 p1;
by rand;
pct=round(100*sae/total,.1);
rand=left(trim(rand))|| '(N=' || compress(total) || ')';
if pct=. or pct < 0.5 then delete;
run;
goptions reset=all device=gif gsfname=psnameb
ftext=swissb;

filename psnameb 'c:\radar\aerad01.gif';
title1 j=c h=14pt '(%) Comparison of On-Therapy Gastrointestinal Adverse Events By

Treatment';
*footnote1 j=l h=12pt 'Note: Gastrointestinal AE of Concern are: Nausea, Abdominal

Pain, Diarrhea, and Vomiting';
footnote1 j=l h=10pt "USER ID:/aerad01.sas &sysdate &systime";
axis1 value=(h=0.9) width=2 label=(h=10pt) c=cx0000cc;
axis2 label=(h=9pt) c=cx0000cc;
axis3 value=(h=0.9) width=3 label=(h=11pt) c=cx000033;

proc gradar data=final;
chart pref/freq=pct

cstars=(aqua red)
staraxis=(axis2,axis2 ,axis3,axis2 ,axis3,axis2 ,
axis1,axis2, axis1,axis2 ,axis2 ,axis1 , axis1,axis2 ,axis1)
starfill=(solid empty)
starcircles=(.25 .75)
wstars=(1 3)
cstarcircles=black
overlay=rand; run; quit; run;

Figure 2. Wedge Type of GI AE Display

WEDGE TYPE DISPLAY OF SERIOUS AE DATA

The following SAS sample code produces wedge type of serious AE (SAE) display.

/* data preparation steps */
goptions reset=all device=gif gsfname=psnameb rotate=landscape gsfmode=replace
ftext=swissb;

filename psnameb 'c:\radar\aerad045.gif';
title1 j=c h=14pt '(%) Comparison of On-Therapy Serious Adverse Events By

Treatment';

footnote1 j=l h=10pt "USER ID:/aerad041.sas &sysdate &systime";
axis2 value=(h=0.9 c=cx000099) width=2.5 label=(h=10pt c=cx000099) c=cx000099;
axis1 label=(h=9pt) width=1.8 c=cx00cc00;
axis3 value=(h=0.9 c=cxff0000) width=3.3 label=(h=11pt c=cxff0000) c=cxff0000;

proc gradar data=final;
chart body1/freq=pct

spklabel=none
cstars=(aqua red)
staraxis=(axis1,axis1 ,axis1,axis1 ,axis2,axis3 ,
axis1,axis3, axis2,axis3 ,axis2 ,axis1 , axis1,axis2 ,axis3)
starfill=(solid empty)
starcircles=(.25 .75)
wstars=(1 3)
cstarcircles=black
overlay=rand; run; quit;

Figure 3. Wedge Type of Serious AE Display

WEDGE TYPE OF QTC DATA DISPLAY

The following SAS sample code produces wedge type of QTc ECG display.

libname sb 'c:\gradar';
data ecg;
set sb.ecg;
if vis_id=10.1 then delete;
if parm_c='QTCF' and ses_id=1 and (3 < vis_id < 11);

keep rand pid visit vis_id value base_chg ;

proc sort;by rand vis_id visit;
proc summary data=ecg;
by rand vis_id visit;
var base_chg;
output out=p1 mean=mchg;

data final;
set p1;
mchg = (mchg * 10) + 50;
va=substr(visit,7,2);
if substr(visit,1,1)='V' and va ^='10' then do;
visit='VISIT' || ' ' || compress(va);

end;
run;

data anno;
length function $8. style $8.;
xsys='3'; ysys='3';when='a';style='swissb';

size=1.1;
x=46.8;
y=63.3;
function='MOVE'; output;
text='-4';
function='SYMBOL'; output;
x=46.8;
y=40.4;
function='MOVE'; output;

text='-4';
function='SYMBOL'; output;

goptions reset=all device=gif gsfname=gifa
ftext=swissb ;

filename gifa 'c:\gradar\aerad02.gif';
title1 j=c h=14pt 'Comparison of QTC(msec) Mean Change from Baseline By

Treatment';
footnote1 j=l h=10pt "USER ID:/aerad02.sas &sysdate &systime";
axis1 major=(n=3) minor=(n=3) value=(h=0.1 c=white)

width=2 label=(h=11pt) c=cx0000cc;

proc gradar data=final;
chart visit/freq=mchg

annotate=anno
cstars=(aqua red)
staraxis=(axis1, axis1 ,axis1,axis1 ,axis1,axis1 ,
axis1,axis1, axis1, axis1)
starfill=(solid empty)
starcircles=(.25 .75)
wstars=(1 3)
cstarcircles=black
overlay=rand;

note move=(47.5,17)pct h=11pt '0';
note move=(45,74)pct h=11pt '-2';
note move=(45,27.5)pct h=11pt '-2';

PostersNESUG 18



7

note move=(47.5,85)pct h=11pt '0'; run; quit;

Figure 4. Wedge Type of QTc ECG Display

OUTPUT FROM ACTIVEX DRIVER

The ActiveX device driver enables the following drill-down menu for viewer to take interaction through menu. The functionality
of the menu items provided are shown in Figure 5.

Figure 5. Drill-Down Menu Provides Different Menu Item Functionalities

The following SAS sample code illustrates the construction of data tips.

libname sb 'c:\gradar';
data ecg;
set sb.ecg;
if vis_id=10.1 then delete;
if parm_c='QTCF' and ses_id=1 and (3 < vis_id < 11);
label base_chg='Mean+5';

keep rand pid visit vis_id value base_chg ;

proc sort;by rand vis_id visit;
proc summary data=ecg;
by rand vis_id visit;
var base_chg;
output out=p1 mean=mchg n=n max=max min=min median=median

mean=mean stderr=stderr p99=p99 p95=p95 q1=q1 q3=q3;

data anno;
length function $8. style $8.;
xsys='2'; ysys='3';when='a';style='swissb';

size=1.1;
x=46.8;
y=63.3;
function='MOVE'; output;

text='-4';
function='SYMBOL'; output;
x=46.8;
y=40.4;
function='MOVE'; output;

text='-4';
function='SYMBOL'; output;

data final;
set p1;
length tip $1022. rand1 $6. ;
mchg = mchg + 5;
va=substr(visit,7,2);
if substr(visit,1,1)='V' and va ^='10' then do;
visit='VISIT' || ' ' || compress(va);

end;
/* data tips statement to construct the information shown in data text box */

tip='ALT="N: ' || compress(round(n,1.)) || '0D'x ||
'Mean: ' || compress(round(mean,.001)) || '0D'x ||
'Max: ' || compress(round(max,.001)) || '0D'x ||
'Min: ' || compress(round(min,.001)) || '0D'x ||
'Median: ' || compress(round(median,.001)) || '0D'x ||

'STDERR: ' || compress(round(stderr,.001)) || '0D'x ||
'P99: ' || compress(round(p99,.001)) || '0D'x ||

'P95: ' || compress(round(p95,.001)) || '0D'x ||
'Q1: ' || compress(round(q1,.001)) || '0D'x ||
'Q3: ' || compress(round(q3,.001)) ||
'"';

rand1=trim(left(rand));

label rand1='Treatment' ;
run;

filename odsout "c:\gradar";
ods html body="aerad03.html" path=odsout;
goptions reset=all device=activex ftext=swissb ;

title1 j=l h=10pt 'Protocol: XYZ12345 Study 099';
title2 j=l h=10pt 'Population: Intent-to-Treat';

title3 j=c h=13pt 'Comparison of QTC(msec) Mean Change from Baseline By
Treatment';

footnote1 j=l h=10pt "USER ID:/aerad03.sas &sysdate &systime";
axis1 value=(h=0.1 c=white)

width=3 label=(f=swissb h=11pt) c=cx0000cc;
proc gradar data=final;
chart visit/freq=mchg

annotate=anno
cstars=(blue red)
staraxis=(axis1, axis1 ,axis1,axis1 ,axis1,axis1 ,
axis1,axis1, axis1, axis1)
starfill=(solid solid)
starcircles=(.25 .75)
wstars=(1 3)
cstarcircles=black
html=tip
name='aerad03.gif'
overlay=rand1;

note move=(47.5,17)pct h=11pt '0';
note move=(45,74)pct h=11pt '-2';
note move=(45,27.5)pct h=11pt '-2';
note move=(47.5,85)pct h=11pt '0';

run;
quit;
ods _all_ close;

Figure 6 shows the QTc ECG display from ActiveX device with data tips feature.

Figure 6. Output from ActiveX Control Device with Data Tips

Figure 7 shows the QTc ECG display from ActiveX device with drill-down menu feature. You can use the menu to change the
output appearance. Figure 8 shows the output from modification of graph properties.

Figure 7. QTc Display by Visit with Drill-Down Menu

Figure 8. Using Menu to Change Output Appearance and Graph Properties

LFT LAB DATA DISPLAY WITH ACTXIMG DRIVER

The Liver Function Test (LFT) data are part of clinical laboratory data collection to test the liver function. The LFT parameters
are: Alanine Aminotransferase (ALAT), Aspartate Aminotransferase (ASAT), Total Bilirubin (BILTOT), and Alkaline
Phosphatase (ALKPH). The spoke type of radar chart with actximg driver is selected for LFT LAB Display. The sample SAS
code and output are shown as follows:

filename odsout "c:\gradar";
ods html body="aerad07.html" path=odsout;
goptions reset=all device=actximg rotate=landscape gsfmode=replace

ftext=swissb ;
title1 j=l h=10pt 'Protocol: XYZ12345 Study 099';
title2 j=l h=10pt 'Population: Intent-to-Treat';
title3 j=c h=13pt 'Change from Baseline of Lab Test By Treatment and Visit';
footnote1 j=l h=10pt "USER ID:/aerad07.sas &sysdate &systime";
axis1 value=(h=10pt) width=3 label=(f=swissb h=11pt) c=cx0000cc;

proc gradar data=final;
by group;
chart visit/freq=mean

across=parm_cc
down=rand
startype=spoke
ncols=8
nrows=2
cframe=cx66ffff
cframeside=cx9999ff
cframetop=cx9999ff
intertile=1
wspokes=2
starlegend=clock
html=tip
name='aedar07.gif'
description=" "; run; quit;

ods _all_ close;

Figure 9. Spoke Type of LFT Multiple Panels Display with Data Tips

OUTPUT FROM JAVAMETA DRIVER

Figure 10 is a sample output from javameta driver.

PostersNESUG 18



12

Figure 10. Spoke Type of LFT Multiple Panels Display from Javameta Driver

ABPM DATA DISPLAY

Ambulatory Blood Pressure (ABP) monitoring is widely used in clinical practice. It is a monitoring measurement of blood
pressure in a 24-hour period and gives a better prediction of clinical outcome than clinical or casual blood pressure
measurements. Normal blood pressure values for adults are < 135/85 mmHg daytime, < 120/75 mmHg night-time, and <
130/80 mmHg for 24 hours.

The javameta driver is used to create a polar chart for ABPM data shown in Figure 11.

Figure 11. Polar Chart of ABPM Data Display

CONCLUSIONS

Graphic output is an important means to convey the information. The nonparametric methods provide a good starting point for
clinical data analyses. The GRADAR procedure provides a new look to the clinical data.

This paper provides some sample SAS code for graphical displays from GRADAR procedure. Sample displays shown in this
paper not only helped with clinical study results interpretation, but also helped the QC tasks of checking the data, tables, and
clinical reports. In conclusion:

� The procedure GRADAR provides the new chart presentations, such as radar chart, star chart, and polar
chart for displaying data.

� GRADAR supports devices such as ActiveX, actximg and javameta for creating interactive graphs.
� The feature of ‘data tips’ allows user to put relevant information in pop-up text box. It is a useful feature to

pack additional information in graphical presentation.

